# Nelinearna regresija

In[1]:=
```
SetOptions[ListPlot, ImageSize → 6 × 72, Frame → True,
  GridLinesStyle → Directive[Gray, Dashed],
  Method → {"GridLinesInFront" → True},
  PlotStyle → Directive[Thickness[Medium], PointSize[Large], Black],
  TicksStyle → Thickness[Medium], AxesStyle → Thickness[Medium],
  LabelStyle → Directive[FontFamily → "Helvetica", FontSize → 12], Axes → False];
SetOptions[Plot, ImageSize → 6 × 72, Frame → True,
  GridLinesStyle → Directive[Gray, Dashed],
  Method → {"GridLinesInFront" → True},
  PlotStyle → Directive[Thickness[Large], PointSize[Large], Red],
  TicksStyle → Thickness[Medium], AxesStyle → Thickness[Medium],
  LabelStyle → Directive[FontFamily → "Helvetica", FontSize → 12], Axes → False];
SetOptions[ContourPlot, ImageSize → 4 × 72, Frame → True,
  GridLinesStyle → Directive[Gray, Dashed], Method → {"GridLinesInFront" → True},
  TicksStyle → Thickness[Medium], AxesStyle → Thickness[Medium],
  LabelStyle → Directive[FontFamily → "Helvetica", FontSize → 12], Axes → False];
```

In[4]:=
```
data =
  Table[{t, 0.5 Sin[0.8 t] + 0.2 Sin[1.2 t] + Random[NormalDistribution[0, 0.1]]},
   {t, -20, 20, 0.1}];
```

In[5]:=
```
pdata = ListPlot[data, PlotRange → All, FrameLabel → {"t (s)", "x (m)"}]
```

Out[5]=




In[6]:=
```
f1[a_, ω_, t_] := a Sin[ω t]
chi21[a_, ω_] :=
 Sum[(data[[i, 2]] - f1[a, ω, data[[i, 1]]])^2, {i, Length[data]}] / Length[data]
contour1 = ContourPlot[Log[chi21[a, ω]], {a, 0.05, 0.75},
   {ω, 0.6, 1.4}, Contours → 20, MaxRecursion → 0,
   PlotPoints → 40, ColorFunction → ColorData["TemperatureMap"],
   AspectRatio → 2, FrameLabel → {"x0 (m)", "ω (s-1)"}];
```

In[19]:=
```
Manipulate[
 Module[{fit, eval, p1, p2},
  eval = {};
  fit = NonlinearModelFit[data, f1[a, ω, t], {{a, 0.4}, {ω, ω0}},
    t, EvaluationMonitor :→ AppendTo[eval, {a, ω, chi21[a, ω]}]];
  p1 = Show[contour1, Graphics[{Red, Thickness[Large],
      Line[eval[[1 ;; n, 1 ;; 2]]]}]];
  p2 = Show[pdata, Plot[f1[eval[[n, 1]], eval[[n, 2]], t], {t, -20, 20}]];
  Row[{
    p1,
    "  ",
    Column[{
      StyleForm[ToString[TraditionalForm[x0 Sin[ω t]]],
       FontSize → 20, FontFamily → "Helvetica", FontColor → Red],
      p2,
      StyleForm[
       ToString[TraditionalForm[χ^2 / N]] <> " = " <> ToString[eval[[n, 3]]],
       FontSize → 20, FontFamily → "Helvetica", FontColor → Red],
      StyleForm["Korelacijska matrika v minimumu:", FontColor → Red],
      StyleForm[TableForm[fit["CorrelationMatrix"],
        TableHeadings → {{"x0", "ω"}, {"x0", "ω"}}], FontColor → Red]
     }, Center, Spacings → 2]
   }]
  ],
 {ω0, {0.93, 1.11}}, {n, Range[1, 10]}
]
```



Out[19]=




$x_0 \sin(t\,\omega)$




$$\frac{\chi^2}{N} = 0.0290324$$

Korelacijska matrika v minimumu:

| | $x_0$ | $\omega$ |
|---|---|---|
| $x_0$ | 1. | 0.03758 |
| $\omega$ | 0.03758 | 1. |

In[10]:=

```
ω0 = 0.8;
t1 = 40;
t2 = 60;
f2[ω_, λ_, t_] := 0.5 Sin[ω t + Tanh[λ t]]
chi22[ω_, λ_] :=
 Sum[(data[[i, 2]] - f2[ω, λ, data[[i, 1]]])^2, {i, Length[data]}] / Length[data]
contour2 = ContourPlot[Log[chi22[ω, λ]], {ω, 0.6, 1},
   {λ, -0.2, 0.2}, Contours → 20, MaxRecursion → 0,
   PlotPoints → 40, ColorFunction → ColorData["TemperatureMap"],
   AspectRatio → 1, FrameLabel → {"ω (s^-1)", "λ (s^-1)"}];
```

In[16]:=

```
Manipulate[
 Module[{fit, fit1, eval, p1, p2},
  eval = {};
  fit = NonlinearModelFit[data, f2[ω, λ, t], {{ω, 0.9}, {λ, 0.05}},
    t, EvaluationMonitor :→ AppendTo[eval, {ω, λ, chi22[ω, λ]}]];
  fit1 = NonlinearModelFit[data, f2[ω, 0, t], {{ω, 0.9}}, t];
  p1 = Show[contour2,
    Graphics[{Red, Thickness[Large], Line[eval[[1 ;; n, 1 ;; 2]]]}]];
  p2 = Show[pdata, Plot[fit1[t], {t, -20, 20},
     PlotStyle → Directive[Blue, Thickness[Large]]],
    Plot[f2[eval[[n, 1]], eval[[n, 2]], t], {t, -20, 20}]];
  Row[{
    p1,
    "  ",
    Column[{
      StyleForm[ToString[TraditionalForm[x0 Sin[ω t]]],
       FontSize → 20, FontFamily → "Helvetica", FontColor → Blue],
      StyleForm[ToString[TraditionalForm[x0 Sin[ω t + Tanh[λ t]]]],
       FontSize → 20, FontFamily → "Helvetica", FontColor → Red],
      p2,
      StyleForm[
       ToString[TraditionalForm[χ^2 / N]] <> " = " <> ToString[eval[[n, 3]]],
       FontSize → 20, FontFamily → "Helvetica", FontColor → Red],
      StyleForm["Korelacijska matrika v minimumu:", FontColor → Red],
      StyleForm[TableForm[fit["CorrelationMatrix"],
        TableHeadings → {{"ω", "λ"}, {"ω", "λ"}}], FontColor → Red]
     }, Center, Spacings → 2]
   }]
  ],
 {n, Range[1, 10]}
]
```






$$x_0 \sin(t\,\omega)$$

$$x_0 \sin(\tanh(\lambda\, t) + t\,\omega)$$

Out[16]=




$$\frac{\chi^2}{N} = 0.0279961$$

`Korelacijska matrika v minimumu:`

| | $\omega$ | $\lambda$ |
|---|---|---|
| $\omega$ | 1. | -0.914781 |
| $\lambda$ | -0.914781 | 1. |

**1. naloga:** Teorija kemijske kinetike napove za sigmoidno krivuljo iz podatkov *Adrenalin.dat* naslednjo odvisnost $F / F_{max} = c / (a + c)$, kjer pomeni $a$ koncentracijo s polovičnim maksimalnim učinkom. S pomočjo nelinearne regresije določi koeficienta $F_{max}$ in $a$. Rezultate primerjaj z rezultati naloge 7.2.

**2. naloga:** V datoteki *RefractiveIndex.txt* so zbrane meritve lomnega količnika nekega materiala v odvisnosti od valovne dolžine pri različnih temperaturah. Odvisnost lomnega količnika od valovne dolžine opisuje Sellmeierjeva formula

$n(\lambda)^2 - 1 = \sum_{i=1}^{N} \frac{S_i}{1-\left(\frac{\lambda_i}{\lambda}\right)^2}$, kjer so $\lambda_i$ valovne dolžine elektronskih prehodov v opazovanem materialu, $S_i$ pa pripadajoče uteži. Z nelinearno regresijo določi $\lambda_i$ in $S_i$ (a) ob predpostavki, da meritev lahko opišemo z enim samim elektronskim prehodom ($N$=1) in (b) ob predpostavki, da lahko meritev opišemo z dvema takima prehodoma ($N$=2). Nariši temperaturno odvisnost valovnih dolžin $\lambda_i$ in uteži $S_i$.